
Sparks, tags, suffixes and subscripts

MetaPost variables are complicated things!

Introduction
MetaPost inherits almost all of its syntax and internal structures

from METAFONT. Unfortunately, it did not inherit METAFONT’s

documentation.

The implicit assumption is that you should have read the META

FONT book by D.E. Knuth and if you want to know more you should

ask a METAFONT guru for help.

But METAFONT gurus are hard to find these days …

Tokens, sparks and tags

MetaPost has a three kinds of tokens:

1. There are numeric tokens (floating point numbers),

2. string tokens (stuff between double quotes),

3. and symbolic tokens (everything else).

Building tokens

MetaPost does not have the \catcode command of TEX. However, it

does have its own internal list of category codes, and those internal

categories are used to construct symbolic tokens using a fairly short

(but perhaps unexpected) list of rules.

The rules:

□ a space character is ignored;

□ a period character is ignored unless followed by another period

or by a digit;

□ a percent sign skips until the next end of line character

□ a decimal digit or a period followed by a decimal digit starts a

numeric token;
□ a double (ascii) quote starts a string token;
□ left or right parenthesis, a comma, or a semicolon, creates a

symbolic token;
□ anything not matched yet combines with the longest following

sequence in the same class to become a symbolic token.



Tokens 6 / 54

September 19, 2019

Numeric tokens
Once the start of a numeric token has been detected, MetaPost runs

a numeric token scanner.

In the default scaled mode, a numeric token is the expected

combination of digits and a dot.

In the other numbersystem modes, an optional exponent can follow

immediately after.

An exponent specification starts with the letter e or E, followed by

an optional + or -, and then a series of digits. No intervening spaces

are allowed.

String tokens

Once the start of a string token has been detected, MetaPost

gobbles up characters from the current input line until it finds the

matching double quote. The resulting string token consists of those

letters.

Symbolic tokens

The last rule from the rule list is what makes e.g. ‘beginfig’ be a

single token. The actual classes are defined by the following list,

and they highlight some of the oddness of the MetaPost input

language.

Character classes

A–Z _ a–z
< = > : |
‘ ’
+ -
/ * \
! ?
# & @ $
^ ~
[
]
.
{ }

Tokenization effects
The statements above explain the existence of some fairly common

MetaPost symbols such as ‘beginfig’, ‘:=’, ‘..’ and ‘---’.
But it also means:

□ that ‘!?!’ and ‘[[[[’ are valid symbols, which could be defined if

you so desired;

□ that ‘(’, ‘)’, ‘,’, and ‘;’, are always standalone symbols.

□ that there can never be symbolic tokens containing spaces,

percent signs, double quotes, or digits;



Tokens 11 / 54

September 19, 2019

□ that period characters are often (but not always) equivalent to

spaces (in fact, MetaPost usually replaces spaces with periods in

log reports);

□ that numeric tokens are never negative (negative numbers

require two tokens);

□ that string tokens are limited to a single line and never contain

double quotes (those strings need to be created using char).

Sparks and tags

In MetaPost, it is not necessarily the case that a symbolic token is

actually a command for the engine.

To aid in the distinction, tokens that do signify commands are called

sparks. The ones that do not are called tags. Sparks are symbolic

tokens that either refer to primitive operations or are defined to be

macros. All other symbolic tokens are tags, and these are used to

build up variable names.

In a simple statement like

w = 12pt;

there are four symbolic tokens and one numeric token: w, =, 12, pt,
and ;.
The w and pt are tags. The other two symbolic tokens (= and ;) are
sparks. (note that pt is actually a variable name)

Sparks and/or tags

MetaPost primitive operations are separate from their name, so

sparks can be altered:

The let command creates an alias for a primitive operation that is

currently attached to a spark. This creates another spark.

save and type declarations can turn a spark into a tag;

def can alter a spark, giving it a different meaning (as well as

creating new ones).

Variable names
Variable names in MetaPost are not limited to a single symbolic

token. Another simple example is a statement like:

x1 = 12pt;

The x and 1 are two segments that are actually combined into a

single variable name.

So what exactly is a variable name?

A variable name is built up from a tag,
optionally followed by a suffix.

A suffix in turn is either a subscript or another tag, possibly
followed by yet another suffix, and so forth.

A subscript is either a numeric token, or a bracketed numeric

expression (which then should result in a known numeric value).

Surprise quiz

What does this equation mean?

x3ab c[2.1+1] f.4 = 12pt;

Quiz solution

The single variable is named ‘x 3 ab c 3.1 f .4’ (seven segments!)

It has a numeric value of 11.95514 (12 times 0.99626).

Subscripts

A subscript can be an immediate numeric token like 3 and .4 in the

above example, or it can be a bracketed expression like [2.1+1] that

directly results in a numeric value. The brackets are required for

MetaPost to interpret the subscript as an expression.

Surprise quiz 2

What does this equation mean?

x3ab c 2.1+1 f. 4 = 12pt;

Quiz solution
That was an equation of two variables:

x[3]ab.c[2.1] + 1*f[4] = 12pt;

Numeric expressions

Numeric tokens cannot be negative, but the result of a numeric

expression can be negative.

The only requirement is that it has to produce a known numeric

value. The following is allowed (although likely not very useful):

a[- floor uniformdeviate 20 + 5] = 12pt;

Variable name parsing effects

The parsing rules mean that

□ a string token can never be part of a variable name,

□ and neither can any spark,
□ and a variable name never starts with a numeric token nor a

numeric expression.

The restriction on sparks in variable names is a cause of common

errors in MetaPost input.

Declarations
There is no need to pre-declare numeric variables. But numeric is

not the only variable type that MetaPost knows about; the other

types do need to be predeclared (otherwise they default to the

numeric type).

MetaPost handles ten different variable types.

Simple declarations

boolean mybool;
cmykcolor mycolor;
color mycolor;
numeric mynumber;
pair mypair;
path mypath;
pen mypen;
picture mypic;
rgbcolor mycolor;
string mystring;
transform mytransform;

Why the numeric declarator exists …

While numeric variables do not need to be predeclared, the numeric
keyword is still useful. That is because declaration commands

completely wipe out the current meaning of the to-be-declared

object, if it already exists.

Declaration lists
Declaration commands can be followed by lists of items:

path p, q;

The argument to a declaration command is not exactly a variable

name (or even a list of those), it is a bit more complicated than that:

For starters, each element of the argument list is allowed to be a

spark.



Declarations 28 / 54

September 19, 2019

Declaration lists destroy sparks!

Of course, after the declaration has been processed, such sparks will

become tags.
Usually this is not what you want, but MetaPost does not give any

warnings about redefining sparks in this way, so you have to be

careful.

The statement

path path;

is allowed. It will be the last working path declaration in the current

run, though, as it will turn path into a tag!

Collective subscripts

MetaPost insists that all variables whose names are identical except

for subscript values must have the same type. You can not have a1
be a pair and a2 be a color, for example.

To enforce this rule, you can not use numeric tokens in declarations,

only so-called ‘collective subscripts’. The declaration would look

like this:

pair a[];

Collective subscripts are not arrays!

Collective subscripts only tell MetaPost that any variable with a

combined name consisting of a followed by a numeric segment
will be of type pair.

This does not prohibit using a as if it is an array, but it is important

to realise that MetaPost does not actually have the concept of arrays

at all.

Collective subscripts are specific

An important advantage of collective subscript declarations is that

they has no influence on any other variables whose names are not

of the form a plus subscript.
Also, the variable a itself in not affected.

Even a nested set of variable names with each level having a

different type is acceptable:

pair a;
path a[];
color a[]c;

A small warning: do not forget that the statements

path a.path;
color a.color;

are both illegal because they would result in variables names with

sparks in them.

You need something like this instead:

path a.pth;
color a.col;

Internal quantities

Besides user-defined variables, MetaPost also has a number of

internal variables. These are officially called internal quantities.
To keep things simple, all the names of these internal variables are

single symbolic tokens.

Examples are charcode, warningcheck and jobname.

Internal quantities cannot be redeclared

It is not possible to change the type of these internal variables.

If you try anyway, you will end up with a new user-defined variable

that happens to have the name of an internal quantity but is in fact

not related to it at all.

From then on, the internal quantity has become inaccessible from

within your code, even though the variable itself still exists. In

situations where MetaPost needs to use that internal variable, it will

use the value it held before you made it inaccessible.



Internal quantities 35 / 54

September 19, 2019

The newinternal command
There is a command to make new internal quantities: newinternal.
Internal quantities can only receive known values, which means

you can not set their values using linear equations, making

newinternal quantities less useful than normal variables.

However, access to internal quantities is a little bit faster than

normal variables, and that is even true for user-defined ones.

It can be handy to define new internal quantities for numerical

constants.

For example, plain.mp defines eps as:

newinternal eps;
eps := .00049;

The save command
The save command processes a list of sparks and/or tags, saves
their current value, and then converts then into undefined tags. It

also makes every sub-variable that starts with that specific name be

undefined.

The save command operates on individual symbolic tokens, so it

can not be used to save just some sub-part of a segmented variable.

It does not wipe-out and replace the previous variable as a new

declaration would but it makes the tag freshly available locally.

The normal use for save is within a group started with begingroup
and ending with endgroup, like within beginfig …endfig.
The traditional beginfig macro contains the equivalent of this line:

save x,y;

to make sure that any values of type x[] and y[] outside of the

current figure do not have any undue influence, while still saving

them for potential later use.

If you use save twice within a single group, the first value saved

‘wins’.

The interim command
For internal quantities, using save would not work, because the

symbolic token becomes undefined and therefore unassignable.

The argument to interim looks like a normal assignment. The only

difference is that the previous value is restored at the end of the

group.

interim warningcheck := 0;

First saved wins, as with save.

let and def
These commands work much like in TEX. There are some differences,

though.

MetaPost does not have user-controlled macro expansion, and

it handles grouping in a completely different way, so the typical

prefixes like \global and \expanded of TEX do not exist.

The let command
The let command makes one symbolic token be an alias for

another symbolic token.

□ If the token on the left hand side is a tag, then all variables that

start with that tag become undefined

□ if the token on the right hand side is a tag, then the LHS

becomes undefined but the variable(s) on the RHS are left as-is

The let command is meant to provide aliases for sparks that you

plan to redefine.

The def command
As mentioned earlier, def (and its cousins primarydef, secondarydef,
and tertiarydef) produce sparks.
Remember, assuming there is a definition like this:

def up = (0,1) enddef;

then the variable name a.up would become forbidden until the

definition goes out of scope.

Variable definitions
The restriction of def always producing a spark is why there is a

dedicated command for creating macros that are actually tags.
This command is called vardef. In simple uses, use of vardef is

very similar to using def.

def stuff =
fill unitsquare

enddef;

and

vardef stuff =
fill unitsquare

enddef;

look equivalent



Variable definitions 44 / 54

September 19, 2019

But there is a difference in execution: the vardef version actually

expands into:

begingroup
fill unitsquare

endgroup

The extra grouping makes the macro expansion syntactically

equivalent to a normal variable, which is important because it

avoids confusing the MetaPost parser.

Grouping effects

Grouping in MetaPost is a bit unusual in that the begingroup
…endgroup block is not only a list of statements.

It can also be used as an expression, where the last expression

inside the group is taken as the value for the outside expression.

That oddity of grouping is what makes vardef definitions syntactically

equivalent to variables.

Incidentally, it also makes grouping behave similar to an anonymous

function call with one return value.

The extra grouping that is inserted means you cannot do things like

stuff withcolor green;

which makes sense once you realise that vardef is supposed to

equate to a variable. So the vardef needs to be adjusted to

vardef stuff =
unitsquare % earlier 'fill' deleted

enddef;

and used like this

fill stuff withcolor green;

Why vardef is actually useful

First, because vardef defines a new tag instead of a spark, the
symbolic token can still can be used in the middle of an unrelated

compound variable.

The METAFONT book highlights the example of dir. The variable

macro dir is defined as a vardef precisely because that still allows a

pair variable named p5dir.

But there is a more important use of vardef. The heading of a

vardef has a syntax that is a little more elaborate than a normal def.
From plain.mp:

vardef z@#=
(x@#,y@#)

enddef;

This defines the macro z. This definition head makes that the

macro now has a built-in parameter of type suffix that is named @#.
(remember that @# is a single token)

There is a subtle difference between this definition of z and the

more naïve version:

vardef z suffix v =
(x.v,y.v)

enddef;

The special token @# only applies to an immediately following

suffix; the suffix that becomes the argument may not be enclosed

in parentheses, unlike in the second definition. This makes it

exceptionally useful for manipulating sub-variables (like z does).

Finally every vardef, with or without the special @#, also has two

other special implicit arguments that can be used anywhere in the

replacement text.

□ @ returns the last segment of the name of the defined macro

□ #@ returns the complement of that: all segments before the last

one

□ @# returns the (predeclared) suffix to the macro

An example:

vardef p[]dir= (#@dx,#@dy) enddef;

After this definition, p5dir expands into (p5dx,p5dy)
allowing you to write, for example:

p5dir = up;

to define the dx and dy subvariables, and query those values by

if p5dir = up: .... fi

… a lot nicer than manipulating the dx and dy variables ‘manually’.

In definitions like p[]dir, the special token @ is not very useful, but

subscripts can also be vardef macros:

vardef a[] =
if odd @: message("odd")
else: message("even")
fi

enddef;
a1; % prints "odd"
a20; % prints "even"
end.

A final warning about vardef
Because vardef is a macro, it only works as the last typed segment.

After the p[]dir definition, you can not declare another subvariable:

pair p[]dir.target; % WRONG!

This is disallowed, because that set of variables would actually be

inaccessible: the target part will always become a suffix argument.

If you really want to write things like p5dir.target, you have to

extend the definition of p[]dir to include @#, and then process the

target manually within the macro expansion.

I hope this was educational!

Resources:

□ The METAFONT book by D.E. Knuth

□ MetaPost – a user’s manual by John. D. Hobby

And thanks go to my proofreaders Alan Braslau and Hans Hagen.


